


## Customer's Know How:

### Schalen in Flächen konvertieren


Von Herrn Uwe Quitmann von der MAN Turbomaschinen AG stammt dieses praktische Makro, das es erlaubt eine schnelle Geometrierückführung zu betreiben. Von einem bestehenden Schalennetz werden Oberflächen abgeleitet.

Beispiel:


**FE Netz ohne Geometrie**



**Schalenelemente an der Oberfläche erzeugen**



**Makro ablaufen lassen und Keypoints mergen**



## Customer's Know How:

Ausgabe: 03 / 2004

### ANSYS Makro:

```
! 1. Schalenelemente selektieren
! 2. Makro ablaufen lassen
! 3. Keypoints mergen
! konvertiert Schalenelemente in Flaechen
! 18.12.2003
! Quitmann / TD1
! Version 1.1

/uis,replot,off
/uis,msgpop,3
*get,n_e,elem,,count    ! Anzahl der Elemente
*dim,kp_feld,array,4,1 ! Dimensionierung des Feldes mit KP-Nummern
cm,sel_elem,elem      ! Komponente erzeugen (selektierte Elemente)
*vfill,kp_feld,data,0,
nr_kp=1
*get,e_nr_neu,elem,0,num,min
*do,i,1,n_e,1
  esel,s,elem,,e_nr_neu
  nsle,s
  *get,n_ns,node,,count    ! Anzahl der Knoten im selektierten Element
  *get,n_min,elem,e_nr_neu,node,1 ! 1.Knoten im Element
  knode,nr_kp,n_min !Erzeuge Knoten
  kp_feld(1,1)=nr_kp
  *do,j,2,n_ns,1
    *get,n_next,elem,e_nr_neu,node,j
    nr_kp=nr_kp+1
    knode,nr_kp,n_next
    kp_feld(j,1)=nr_kp
  *enddo
  *if,n_ns,eq,3,then
    a,kp_feld(1,1),kp_feld(2,1),kp_feld(3,1)    ! Dreiecksflaeche
    *elseif,n_ns,eq,4
      a,kp_feld(1,1),kp_feld(2,1),kp_feld(3,1),kp_feld(4,1) ! Vierecksflaeche
  *endif
  e_nr_alt=e_nr_neu
  cmsel,s,sel_elem
  *get,e_nr_neu,elem,e_nr_alt,nxth
  nr_kp=nr_kp+1
*enddo
```